

GRU_A_PONTE_Beam188

```

FINISH
/CLEAR
C*****
C***
C*** GRU A PONTE BITRAVE
C***

C*** PARAMETRI
/PREP7
IRT=5000 ! Interasse ruote testata
S=20000 ! Scartamento
SC=2500 ! Scartamento carrello
H1=700 ! altezza trave principale
H12=400 ! altezza trasversale
B1=500 ! larghezza
SP1=8 ! spessore
H2=350 ! altezza testata
B2=200 ! larghezza
SP2=5 ! spessore
IRC=2000 ! interasse ruote carrello
PPP=2*10**5 ! portata gru
ESZ1=2000 ! dimensione elementi travi principali
ESZ2=1000 ! dimensione elementi testate
*ASK,YC1, Posizione carrello lungo la trave in mm, 3000

C***
C*** CALCOLO CARATTERISTICHE SEZIONI
C***
C*** trave principale
C***
JY1=(SP1*(H1-2*SP1)**3)/6+B1*SP1*(H1-SP1)**2/2
JZ1=(SP1*B1**3)/6+(H1-2*SP1)*SP1*H12**2/2
J01=2*SP1*((H1-SP1)*(B1-SP1)**2)/(B1-SP1+H1-SP1)
A1=(H1+B1)*SP1*2
C***
C*** testata
C***
C*** JY2=(SP2*(H2-2*SP2)**3)/6+B2*SP2*(H2-SP2)**2/2
C*** JZ2=(SP2*(B2-2*SP2)**3)/6+H2*SP2*(B2-SP2)**2/2
A2=(H2+B2)*SP2*2
C***
C*** KEYPOINTS
C***
K,1
X1=(IRT-SC)/2-H12
K,2,X1
K,3,IRT-X1
K,4,IRT
Z2=(H1-H2)/2
Y2=B2/2+Z1/TAN(25/180*3.14)
K,9,X1,Y2,-Z2
K,11,X1,YC1,-Z2
K,13,X1,YC1+IRC,-Z2
K,15,X1,S-Y2,-Z2
C***
C*** LINEE
C***
L,1,2
L,2,3
L,3,4
LGEN,2,1,3,,S,,4
L,9,11
L,11,13
L,13,15
LGEN,2,-3,,IRT-2*X1,,1
L,2,9
L,3,10
L,15,6

```

```

L,16,7
C***
C*** ELEMENTI
C***
ET,1,188,,3
sectype,1,beam,asec
secdata,A1,JY1,0,JZ1,,J01,,,,H1,B1
sectype,2,beam,hrec
secdata,B2,H2,SP2,SP2,SP2,SP2
C***R,1,A1,JZ1,JY1,H1,B1
C***R,2,A2,JZ2,JY2,H2,B2
MP,EX,1,210000
C***
C*** trave principale
C***
SECNUM,1
LSEL,,LOC,Y,0.1,S-0.1
ESIZE,ESZ1
LMESH,ALL
C***
C*** testate
C***
C***REAL,2
SECNUM,2
LSEL,,LOC,Y,-1,0.01
LSEL,A,LOC,Y,S-0.01,S+1
ESIZE,ESZ2
LMESH,ALL
LSEL,ALL

C***
C*** VINCOLI
C***
DK,1,UY,0
DK,1,UX,0
DK,1,UZ,0
DK,5,UY,0
DK,5,UX,0
DK,5,UZ,0
DK,4,UY,0
DK,4,UZ,0
DK,8,UY,0
DK,8,UZ,0

C***
C*** CARICHI
C***
C*** carico di lavoro
C***
FK,11,FZ,-PPP/4
FK,11,MY,PPP*H12/8
FK,13,FZ,-PPP/4
FK,13,MY,PPP*H12/8
FK,12,FZ,-PPP/4
FK,12,MY,-PPP*H12/8
FK,14,FZ,-PPP/4
FK,14,MY,-PPP*H12/8
C***
C*** peso proprio
C***
ESEL,,REAL,,1,1
SFBEAM,ALL,1,PRESS,A1*9.81*7.8E-6
ESEL,,REAL,,2,2
SFBEAM,ALL,1,PRESS,A2*9.81*7.8E-6
ESEL,ALL
FINISH
/SOLU
SOLVE

```

```

C***
C*** POST-PROCESSING
C***
/POST1
SET
C***
C*** OPZIONI DI RAPPRESENTAZIONE
C***

C*** scelta oggetti da rappresentare
/PBC,ALL,,1
/PBC,NFOR,,0
/PBC,NMOM,,0
/PBC,RFOR,,0
/PBC,PATH,,0
/PNUM,NODE,1
/PSF,PRESS,NORM,2
C*** scelta punto di vista
/VIEW,,1,-1,1
/ANGLE,,-60,ZS
C*** scelta disposizione finestra/legenda
/PLOPTS,INFO,2
C*** scelta colori
/COLOR,PBAK,OFF
*ASK,IFL,VIDEO A COLORI INVERTITI (<=0 SI, >0 NO):,1
*IF,IFL,LE,0,THEN
  /RGB,INDEX,100,100,100,0
  /RGB,INDEX,80,80,80,13
  /RGB,INDEX,60,60,60,14
  /RGB,INDEX,0,0,0,15
*ELSE
  /RGB,INDEX,0,0,0,0
  /RGB,INDEX,60,60,60,13
  /RGB,INDEX,80,80,80,14
  /RGB,INDEX,100,100,100,15
*ENDIF

C*** struttura con vincoli e carichi
/ESHAPE,1
EPLOT
*ASK,IFL,PREMERE RETURN PER CONTINUARE,0

C***
C*** DIAGRAMMI
C***
C*** deformata
/PNUM,NODE,0
PLDISP,1
*ASK,IFL,PREMERE RETURN PER CONTINUARE,0

C*** caratteristiche di sollecitazione
/PBC,ALL,,0
/PSF,PRESS,NORM,0
ETABLE,NI,SMISC,1          ! estrae il dato "sforzo normale" dal data base Fx
ETABLE,NJ,SMISC,14
ETABLE,TZI,SMISC,5        ! estrae il dato "taglio" dal data base SFz
ETABLE,TZJ,SMISC,18
ETABLE,MYI,SMISC,2        ! estrae il dato "momento flettente" dal data base My
ETABLE,MYJ,SMISC,15
ETABLE,MTI,SMISC,4        ! estrae il dato "momento TORCENTE" dal data base TQ
ETABLE,MTJ,SMISC,17
PLLS,TZI,TZJ              ! rappresenta il taglio z
*ASK,IFL,PREMERE RETURN PER CONTINUARE,0

PLLS,MYI,MYJ              ! rappresenta il momento flettente Y
*ASK,IFL,PREMERE RETURN PER CONTINUARE,0

```

PLLS,MTI,MTJ

GRU_A_PONTE_Beam188
! rappresenta il momento torcente X